
Hoofdstuk 1

Inleiding

1.1 Computeralgebra

Een computeralgebrasysteem is een pakket van programma’s waarmee je op een computer
‘algebra’ kunt doen. In de praktijk moet je het begrip algebra ruim opvatten, maar het gebruik
in deze context is vooral bedoeld om te benadrukken dat we algebraı̈sche manipulaties willen
doen op wiskundige objecten, en niet zozeer numerieke berekeningen. Als synoniem wordt wel
de uitdrukking symbolisch rekenen gebruikt. Om de gedachten te bepalen: in computeralgebra
ligt de nadruk meer op het ‘formele’ rekenen met discrete objecten (zoals gehele getallen, en
breuken, maar vooral abstractere objecten als polynomen, elementen in abstracte groepen,
ringen, enz.) dan op bijvoorbeeld het rekenen met reële of complexe getallen.

In de praktijk hebben de meeste computeralgebrapakketten ook faciliteiten om met (func-
ties van) reële getallen te werken, simpelweg omdat dat in veel toepassingen onontbeerlijk is.
Wij zullen die aspecten ook beslist niet willen negeren; aanvankelijk zullen we het pakket dat
we gaan gebruiken vooral zien als een (zeer) veredelde calculator. Maar hoe verder je komt,
hoe meer behoefte er ook gaat komen aan de meer symbolische kant.

1.2 Wat valt er te leren?

Wanneer je voor het eerst een computeralgebrapakket wilt gebruiken om een wiskundig pro-
bleem op te lossen, doet zich een aantal nieuwe problemen voor.

(i) de keuze van het pakket;

(ii) het opstarten (en afsluiten), en het invoeren van commando’s en data;

(iii) het leren (gebruiken) van de taal van het pakket;

(iv) het vinden van de juiste commando’s en datastrukturen;

(v) het vertalen van je probleem van wiskunde naar de programmeertaal.

We zullen op die vragen in deze inleiding in de aangegeven volgorde kort proberen in te gaan.

1.3 Vertalen

Het probleem om een wiskundige opgave te vertalen naar een voor de computer oplosbare
vraag heeft natuurlijk vele facetten. Om te beginnen zou je eigenlijk al van de vraag af
willen laten hangen welk pakket je kiest, omdat soms voor specifieke vragen speciale software
is ontwikkeld. Is, zoals voor ons, die keuze eenmaal bepaald, dan moet je de wiskundige
objecten in Magma-objecten omzetten, je afvragen op welke manier je probleem in principe
opgelost zou kunnen worden, en dan de juiste commando’s toepassen om dat door Magma uit
te laten voeren. Kortom alle andere vragen uit het rijtje komen aan bod. Bovendien moet je,
als je een antwoord krijgt, dat antwoord weer weten te interpreteren in de context waarmee je
begon. Een belangrijke regel is dat computeralgebrapakketten, vooral bij meer geavanceerde
problemen, wel eens foute of incomplete antwoorden kunnen opleveren. Het is dus van belang
om na te gaan dat het gevonden antwoord zinnig is. Bovendien komt het vaak voor dat je
langer op een antwoord moet wachten dan je lief is. In dat geval verdient het aanbeveling te
zoeken naar een manier om Magma te helpen, door een eenvoudigere methode te kiezen, of
het probleem in stapjes op te lossen zodat je kunt zien waar Magma moeite mee heeft. Soms
zal dan blijken dat je vraag niet goed (genoeg) gesteld was. Het kan ook zijn dat je beter een

6 HOOFDSTUK 1. INLEIDING

numerieke dan een symbolische methode kunt proberen, enzovoorts. Natuurlijk weet je in het
begin voor veel van de moeilijkheden die zich voordoen geen oplossing.

1.4 Doel

De bedoeling is dat je na deze inleidende werkcolleges een aantal eenvoudige taken met Magma

kunt uitvoeren, zoals het doen van sommige standaardberekeningen die in de colleges Lineaire
Algebra of Getallen voorkomen. Bovendien is het de bedoeling dat je tegen die tijd zo goed
de weg kent dat je zonder begeleiding in staat bent steeds ingewikkeldere programma’s te
schrijven.

In de volgende hoofdstukken worden heel veel commando’s alleen maar kort genoemd,
zonder precieze uitleg over het gebruik. De bedoeling is dat de naam je voldoende houvast
geeft om met de online help verder uit te vinden hoe de functie van dienst kan zijn.

1.4.1 Voorbeeld

In het tentamen Lineaire Algebra 1 kwam als opgave voor het oplossen van het lineaire stelsel
vergelijkingen:







x1 + x2 + x3 = 2
−x1 + x2 + 2x3 = −1
x1 + x3 = 0.

Dat is met Magma ook eenvoudig; het probleem is om de juiste commando’s te vinden.

> K := RationalField();

> m := Transpose(Matrix(K, 3, 3, [1, 1, 1, -1, 1, 2, 1, 0, 1]));

> v := Vector(K, [2, -1, 0]);

> s, N := Solution(m, v);

> s;

( 1 2 -1)

> N;

Vector space of degree 3, dimension 0 over Rational Field

> s*m;

( 2 -1 0)

Vervolgens werd gevraagd voor welke waarden van α het stelsel







x1 + x2 + x3 = 2
−x1 + x2 + 2x3 = −1
x1 + αx2 + x3 = 0.

geen oplossingen heeft. Ook daarmee heeft Magma geen moeite:

> F<a> := FunctionField(Rationals());

> m := Transpose(Matrix(F, 3, 3, [1, 1, 1, -1, 1, 2, 1, a, 1]));

> v := Vector(F, [2, -1, 0]);

> s, N := Solution(m, v);

> s;

((5/3*a - 1)/(a - 1) -2/(a - 1) (1/3*a + 1)/(a - 1))

waarbij uit de output duidelijk blijkt dat er alleen geen oplossing is als α = 1.

1.4.2 Voorbeeld

In het Analyseboek wordt bij Hoofdstuk 4 gevraagd een N te bepalen zodat voor alle n ≥ N
geldt

2−n + 3−n + 4−n <
1

365
.

In Magma kan dat bijvoorbeeld zo:

> n := 0;

> while (2^(-n)+3^(-n)+4^(-n) ge 1/365) do

> n := n+1;

> end while;

> n;

9

1.5 Gebruik

Een computeralgebrasysteem bestaat meestal uit meerdere componenten en kan op verschillen-
de manieren gebruikt worden. We zullen hier beginnen met het interactieve gebruik, waarbij je
commando’s aan Magma geeft (door ze in te typen) en antwoorden terugkrijgt. Later, wanneer
wat uitgebreidere programma’s geschreven gaan worden, wil je ook wel met een teksteditor
een programma in een bestand schrijven en Magma op dat bestand loslaten. Dat kan ook wel
interactief (door het bestand in te lezen, zoals we binnenkort zien) of de tekst met knippen-
en-plakken in de interactieve sessie in te voeren.

Voor de omgang met de ingebouwde functies (de intrinsics) en het bewerken van de varia-
belen beschikt een computeralgebrasysteem over een programmeertaal waarin je programma’s
schrijft. De taal van Magma is voornamelijk imperatief, hetgeen ongeveer betekent dat je
waarden uitrekent (met een intrinsic of een zelf geschreven user functie) en die waarden dan
met := toekent aan een variabele (of identifier). In bovenstaand voorbeeld zag je dat gebeu-
ren met de variabele n. De taal kent allerhande controlestructuren (zoals while ... end

while boven), waarvan we er een aantal zullen leren gebruiken. Bovendien moeten we leren
allerhande wiskundige objecten (specifieke datastructuren) in Magma te creëren. De taal en
de datastructuren van Magma zijn met opzet zeer wiskundig van aard. Magma is veel minder
dan bijvoorbeeld Maple geschikt voor gebruik door niet-wiskundigen.

In een interactieve sessie (na het opstarten van Magma, in het volgende hoofdstuk uitge-
breider beschreven, maar gewoonlijk gedaan door magma te typen in een shell), geeft Magma

je steeds een prompt om aan te geven dat je input kunt geven. Meestal is dat het symbool >.
Als Magma aan het rekenen is kan het even duren voor de prompt weer verschijnt; eventuele
input die je dan geeft wordt niet gelezen (behoudens de interrupt <control C>) tot de prompt
weer verschijnt.

1.6 Versies, Omgeving

Magma is beschikbaar op verschillende ‘soorten computers’ (onder besturingssystemen als
Unix en Linux, Windows, en MacOS. In deze handleiding gaan we uit van een Unix-
achtige omgeving; interactie met de omgeving speelt voornamelijk een rol bij het opstarten,
bij het creëren en gebruiken van bestanden voor Magma, en bij de lokatie en toegankelijkheid
van hulpbronnen. Een enkele keer verwijzen we naar omstandigheden die specifiek zijn voor
het gebruik van Magma in Nijmegen. Ook verschijnen er regelmatig nieuwe versies; deze tekst
is gebaseerd op versies 2.10 en 2.11. Je ziet de versie bij opstarten in de ‘banier’.

Hoofdstuk 2

Eerste stappen

2.1 Begin en eind

2.1.1 Begin

Om Magma te activeren kun je vanuit een shell het commando magma uitvoeren, of je kunt
onder de math tools (die je window-manager vast ergens vertoont) Magma selecteren.

Eén manier om Magma te gebruiken, is als een veredelde calculator, waar je het antwoord
op een vraag direkt hoopt terug te krijgen — dat is interactief gebruik. De kunst is dan de
juiste commando’s te vinden.

Bij niet-interactief gebruik schrijf je een kort of lang programma in een file en lees je dat
in. Daartoe kun je het commando load "filenaam" gebruiken. Met de quotes kun je Magma

(onder Unix) niet alleen een naam van een file in de huidige directory geven, maar ook een
heel pad, op de gebruikelijke manier onder Unix.

Bij het opstarten kun je door middel van magma -s filenaam in plaats van magma direct
een ‘start-up’ file in laten lezen (met bijvoorbeeld persoonlijk instellingen, definities, etc.).

2.1.2 Algemeenheden

De meeste programma’s in Magma bestaan uit commando’s die de toekenning van een waarde
aan een variabele inhouden; dit gebeurt door middel van := als in x := 4; elke stap wordt
door een ; afgesloten. Stappen mogen meerdere regels beslaan, maar er mogen ook meerdere
stappen op 1 regel staan.

Een ander veel voorkomende stap is om Magma te vragen de waarde van een variabele te
tonen; dat kan (voor de variabele ‘x’) simpelweg met x; of met print x;.

In een interactieve sessie kun je met de ’pijltjes’ toetsen over de regel lopen en veranderingen
aanbrengen; met een pijltje omhoog ‘lopen’ haalt eerdere regels terug. Als je dit midden in
een regel doet loop je terug door regels die net zo beginnen als de huidige regel.

Met SetLogFile kun je alle commando’s en uitvoer in een interactieve sessie naar een
bestand wegschrijven. Dus met SetLogFile("mijnLog") schrijf je alles weg naar het einde van
een bestand ‘mijnLog’ dat zonodig eerst aangemaakt wordt. Met UnsetLogFile() beëindig
je het wegschrijven.

2.1.3 Eind

Je kunt de interactieve Magma-sessie afsluiten door quit; te typen, of exit;. Een alternatief
is <Ctrl>-D, of, wanneer Magma in het midden van een berekening is, tweemaal snel achter
elkaar <Ctrl>-C. In het uiterste geval kan het Magma-proces ook extern ge‘kill’d worden.

Met een enkele <Ctrl>-C onderbreek je de berekening en komt er weer een prompt.

magma -s := print exit quit load SetLogFile UnsetLogFile <Ctrl>-C <Ctrl>-D

9

2.2 De filosofie van Magma

Belangrijk element van de filosofie in Magma is de getypeerdheid van objecten: elk object heeft
een uniek bepaald type (dat je vrijwel altijd te zien kunt krijgen met de functie Type), dat
bepaald wordt bij de creatie van het object, en slechts in uitzonderlijke omstandigheden kan
veranderen. Zo kun je een veelterm 3 · x2 − 1 in Magma niet zonder meer letterlijk intypen,
omdat je tevoren eerst moet aangeven wat x eigenlijk is, en waar het object veelterm leeft.
Heel vaak moet je daarom eerst een wiskundige structuur definiëren waarin het object zijn
plaats kan krijgen; in dit voorbeeld zul je eerst de ring moeten definiëren waarin de veelterm
leeft; dit moederobject (meestal met Parent) terug te vinden), bepaald ook welke operaties
op het element toegestaan zijn.

Een belangrijke uitzondering op de onveranderlijkheid van het type van een object bestaat
in typecoërcie, waar een gedwongen verandering van type (en moederobject) plaatsvindt. Soms
gebeurt dit automatisch in Magma (bijvoorbeeld bij de creatie van ‘verzamelstructuren’), soms
wil de gebruiker het zelf doen. Voor dat laatste is de operator ! ingevoerd. Als simpel
voorbeeld verandert in Integers() ! (4/2); het type van het resultaat van de deling van
4 door 2 van ‘rationaal getal’ in ‘geheel getal’.

2.3 Documentatie

Alvorens te beginnen met de beschrijving van het systeem is het wellicht nuttig een aantal tips
te geven voor het verkrijgen van verdere hulp buiten deze handleiding om.

Er is een Handbook of Magma-functions beschikbaar bij Magma waarin (bijna) alle func-
tionaliteit van Magma beschreven staat. Dit is meer een encyclopedisch werk (ongeveer 3800
pagina’s en een index) dan een tekst waaruit je het gebruik van Magma kunt leren. Er zijn
diverse manieren om toegang te krijgen tot dit Handbook.

(1) De tekstbestanden zijn in diverse vormen beschikbaar in subdirectories van de directory
waar Magma zich bevindt. In Nijmegen is dat in /vol/magma/pdf bijvoorbeeld, voor
hb.pdf (er zijn ook .dvi, en .ps versies), die je met de acrobat lezer met acroread (of
met xdvi, en ghostview voor de andere versies) kunt lezen. Omdat de file zo groot is is
er ook een versie (hb1.pdf tot en met hb8.pdf) die in stukken is gesplitst; in die versie
staat het meest relevante materiaal voor deze Handleiding in hb1 (inclusief de index) en
hb4.

(2) Dezelfde bestanden kunnen ook met een web-browser bekeken worden; met het comman-
de magmahelp in een shell start je een browser op die begint op de introductiepagina van
deze faciliteit. De .html bestanden zijn te vinden in /vol/magma/htmlhelp/.

(3) Wanneer je eenmaal in Magma bezig bent kun je, wanneer je de ‘prompt’ ziet ?? typen
om in de Magma help-browser terecht te komen. Dit laat je binnen de Magma-sessie zelf
navigeren in de tekstbestanden waaruit ook de .pdf files en htmlhelp zijn opgebouwd.
De help-browser legt zelf uit hoe dat moet (de eerste stap is het verzoek om ?? te
laten volgen door help. Je kunt in de bestanden ongeveer navigeren al sin een Unix

omgeving.). Je verlaat de help-browser door quit te typen (zonder ;).

(4) Er zijn ook enkele mogelijkheden om wat gerichter naar informatie te zoeken, vooral
wanneer je al ongeveer weet wat je zoekt, binnen een lopende Magma-sessie.

(i) Wanneer je van het bestaan van een functie in Magma op de hoogte bent, maar
bijvoorbeeld informatie over het preciese gebruik of de argumenten wilt hebben,
kunnen de naam typen gevolgd door ;. Zo zal

Factorization;

je (misschien niet direct heel begrijpelijk) uitleggen welke argumenten precies zijn
toegestaan en wat je als resultaat krijgt. De eerste regel

(<RngIntElt> n) -> RngIntEltFact, RngIntElt, SeqEnum

zegt bijvoorbeeld dat je een RngIntElt (dat is een geheel getal) n als argument
mag geven, en als resultaat drie objecten terugkrijgt. De uitleg

Attempt to find the prime factorization of the value of n; the

factorization is returned, together with the appropriate unit and the

sequence of composite factors which could not be factored.

vertelt wat die drie objecten zijn.

(ii) Soortgelijke informatie, maar geordend volgens de inhoud van de help-browser, krijg
je door ?Factorization te typen; je krijgt dan een lijst van (in dit geval 31) plaatsen
in de documentatie waar uitleg over de functie Factorization te vinden is, en ook
of het om een Sectie uit het Handbook gaat, om de beschrijving van een Intrinsic
functie, of een Example: zo krijg je na

?Factorization

onder 30 andere

17 I /magma/ring-field-algebra/integer/creation/conversion/\

FactorizationToInteger

te zien, hetgeen betekent dat je nu met

?17

uitleg over het gebruik van de gerelateerde functie FactorizationToInteger kunt
krijgen.

(iii) Voor beide voorgaande opties moet je al weten naar welke naam van een functie je
zoekt. Om je daarbij te helpen kun je als volgt zoeken naar het bestaan van namen
voor functies in Magma. Wanneer je het begin van een mogelijke functienaam typt
(die namen beginnen meestal met een Hoofdletter) en vervolgens de Tabulator toets
<Tab> dan geeft Magma een lijst van alle functienamen die zo beginnen. Als er maar
precies 1 mogelijkheid is om een bestaande functienaam te krijgen vult Magma

het vervolg aan en wacht dan eventueel op een nieuwe <Tab> om alle mogelijke
verlengingen te tonen. Zo leidt

> Fac<Tab>

tot

> Factor

en vervolgens

> Factor<Tab>

tot een lijst van ongeveer 20 functienamen die met Factor beginnen. Je wordt
geacht vervolgens 1 toegestane mogelijkheid te typen (of een deel, weer gevolgd
door <Tab>).

(iv) Een andere manier om mogelijke functie te vinden is door ListSignatures te
gebruiken. Zo geeft ListSignatures(RngIntElt); je een (gigantische!) lijst van
alle functies in Magma waar een geheel getal als (één van de) argument(en) is
toegestaan.

Opgave 1. Welke andere mogelijkheden beginnen met List?

<Tab> ListSignatures

Hoofdstuk 3

Eenvoudige datatypen

3.1 Getallen

Diverse soorten getallen zijn in Magma uitzonderlijke objecten omdat je ze letterlijk kunt
creëren zonder eerst hun ‘moederstructuur’ aan te geven. Dat is niet alleen omdat je ergens
moet kunnen beginnen met objecten te creëren, maar ook omdat er bij getallen meestal geen
enkel misverstand bestaat over wat er bedoeld wordt. Het is dan ook gemakkelijk om gehele
getallen, breuken, reële, en complexe getallen te maken en ermee te rekenen, met de operatoren
+, -, *, /, ^ die de gebruikelijke betekenis hebben.

3.1.1 Gehele getallen

Gehele getallen worden simpelweg letterlijk ingetypt, zoals bijvoorbeeld -13. Er is geen spe-
ciaal type van natuurlijke getallen. Soms is het nodig om Z (als ring van gehele getallen)
expliciet in handen te hebben; daarvoor is Integers() geschikt.

Met mod kun je rekenen modulo een getal m.
Belangrijk voor latere constructies: de rijtjes gehele getallen van a tot b met stapjes ter

grootte c maak je met [ a..b by c]. Als je de stapgrootte weglaat wordt deze op 1 gezet.

Opgave 2. Creëer de gehele getallen −11, 0, 11 in Magma en wijs ze aan de variabelen x, y, z toe.

Wat is de naam die Magma voor het type van deze objecten gebruikt, en wat is de naam van de

moederstructuur?

Opgave 3. Op hoeveel verschillende manieren kun je legale uitdrukkingen van de vorm a ◦ b � c als
a, b, c ∈ {2, 3, 4} en ◦, � beide één van bovenstaande operatoren mogen zijn? Geef twee antwoorden:

(i) wanneer a, b, c en ◦, � verschillend moeten zijn;
(ii) wanneer getallen en operaties hetzelfde mogen zijn.

(iii) Probeer een flink aantal van de mogelijkheden uit de vorige opgave uit in Magma, en zie of de

uitkomsten altijd zijn wat je ervan verwacht (bijvoorbeeld, wat is 23
4

?

Opgave 4. Probeer nogmaals een aantal mogelijkheden uit, en voorspel van welk type het resultaat

zal zijn; gebruik positieve en negatieve getallen (en 0) in teller en exponent).

Opgave 5. Wat is het resultaat van −11 mod 4 in Magma, en van welk type is het?

3.1.2 Rationale getallen

Rationale getallen maak je eenvoudig met /, als in 4/8. Teller en noemer (na simplificatie!)
vind je terug met Numerator en Denominator. Het lichaam der rationale getallen maak je met
RationalField().

+ - * / ^ Integers mod [ .. by ] RationalField Numerator Denominator

13

3.1.3 Reële getallen

Reële getallen kun je invoeren door decimalen 0-9 en een decimale . te gebruiken; je krijgt ze
natuurlijk ook als resultaat van bepaalde functies, zoals logaritmen, trigonometrische functies,
enzovoorts. Met RealField() wordt het lichaam van de reële getallen expliciet gemaakt.

Sommige constanten hebben hun eigen creatiefunctie, zoals π, met Pi.
Er wordt getracht op een intelligente manier om te gaan met het aantal decimalen dat

correct is voor een reëel getal; zo zal worden onthouden dat een ‘geheel reëel getal’ met
‘oneindige’ precisie bekend is; in het algemeen zijn natuurlijk maar eindig veel (vaak 28)
decimalen correct bekend. Het vereist enige oefening om met de precisie van reële getallen om
te gaan; de functie Precision geeft het aantal decimalen.

Opgave 6. Creëer het reële getal −3 op drie verschillende manieren: letterlijk, door coërcie van een

geheel getal, en door coërcie van een rationaal getal.

Opgave 7. Vergelijk de typen van bovenstaande reële getallen en hun precisie.

Opgave 8. Maak π2/4.

Opgave 9. Raad de naam voor de cosinus functie en haar inverse in Magma, en bereken de cosinus

van 45 graden en cos−1(1/2). Hoe zit het met de precisie van de uitkomsten?

Opgave 10. Creëer het getal e, de basis van de natuurlijke logaritmen.

Opgave 11. Bereken de logaritmen van e met grondtal e en met grondtal 10.

3.1.4 Complexe getallen

De complexe getallen vormen het eerste voorbeeld waar we pas elementen kunnen definiëren
nadat de moederstructuur bestaat. Dat is omdat in Magma niet de letter i gereserveerd is
voor

√
−1. De manier om dit te doen is door middel van C<ii> := ComplexField(). Dit

creëert tegelijkertijd het lichaam der complexe getallen (en kent dit object toe als waarde aan
de variabele C) maar ook de voortbrenger

√
−1, die hier aan de variabele ii wordt toegekend.

Opgave 12. Creëer de complexe getallen zoals hierboven, en maak de complexe getallen 3 +
√
−1 en

π/(1 +
√
−1).

Opgave 13. Bepaal de waarde, het type, en de moederstructuur van
√
−1

2

wanneer die als in de vorige

opgave is gemaakt.

Opgave 14. Laat zien dat het niet werkelijk noodzakelijk is om het lichaam C eerst te creëren: je

kunt de wortel uit het gehele getal −1 trekken, en de waarde aan de variabele i toekennen. Nu kun je

complexe getallen hiermee maken. Laat dat zien. Wat valt je op als je zo’n complex getal afdrukt?

Opgave 15. Er zijn in Magma tenminste 4 functies waarmee je de wortel uit een getal kunt berekenen:

Root, Sqrt, Iroot, en Isqrt. Vind uit wat de vereiste input is om
√

m voor verschillende waarden en

typen van m uit te rekenen en bekijk ook het type van de output.

RealField Pi Precision ComplexField Root Iroot Sqrt Isqrt

3.2 Vergelijkingen and Boolese waarden

Gelijkheden en ongelijkheden maak je met eq, ne, gt, ge, lt, le, voor ‘is gelijk’, ‘is ongelijk’,
‘is groter dan’, ‘is groter dan of gelijk aan’, ‘is kleiner dan’, en ‘is kleiner dan of gelijk aan’.

Het resultaat van zo’n vergelijking is een Boolese waarde, die je ook met true of false
kunt maken. De operatoren and, or, not, kunnen op voor de hand liggende wijze worden
gebruikt.

De structuur waarin Boolese waarden thuishoren is Booleans().

Opgave 16. Vraag Magma of de waarheidswaarden van 5 < 4 en de ontkenning van ‘waar’ hetzelfde

zijn.

Opgave 17. Wat is het type van true?

3.3 Karakters, woorden, commentaar

Soms is het nuttig over ‘strings’ van letters te kunnen beschikken. De structuur Strings()

herbergt deze objecten; de elementen kunnen gemaakt worden door de karakters tussen dub-
bele aanhalingstekens te plaatsen, dus bijvoorbeeld w := "woord";. Met w[i] kan (voor een
natuurlijk getal i) dan de i-de letter verkregen worden. Met * kunnen strings acter elkaar
geplakt worden; met ˆ4 kunnen (bijvoorbeeld) 4 copiën aan elkaar geregen worden.

Als speciale symbolen moeten de dubbele quotes en de backslash voorafgegaan worden
door een backslash om ze als karakter te krijgen, dus \" en \\. Verder betekent \n een ‘nieuwe
regel’ en \t een ‘tab’.

Regels beginnend met een dubbele slash // worden door Magma als commentaar opgevat,
en verder genegeerd.

Opgave 18. Laat Magma de regel
welkom bij Magma!

afdrukken.

Opgave 19. Laat Magma nu de regels
"Hartelijk welkom"

nieuwe Magmagebruiker

afdrukken.

Opgave 20. Verklaar het verschil in uitkomst tussen de twee volgende Magma statements:

("73" * "9" * "42" eq "7" * "3942") eq (73 * 9 * 42 eq 7* 3942);

("73" * "9" * "41" eq "7" * "3941") eq (73 * 9 * 41 eq 7* 3941);

3.4 Coërcie

Een belangrijk karakter in Magma is het uitroepteken !. Het wordt vaak gebruikt om het type
van een object te veranderen: met R ! x wordt gepoogd om het element x op te vatten als
element van de structuur R. Zo zal t := ComplexField() ! -2; leiden tot de creatie van
een complex getal t dat gelijk is aan −2.

Opgave 21. Coërcie wordt vaak gebruikt wanneer een operatie niet van toepassing op een bepaald

type object, maar wel wanneer we het interpreteren als een ander type. Laat zien hoe je de priemge-

talontbinding van 1155/7 kan vinden (met Factorization).

Opgave 22. Voor het uitrekenen van een faculteit bestaat de functie Factorial en kan ! niet gebruikt

worden. Bepaal 5! als reëel getal.

Booleans() true false eq ne gt lt ge le Strings() " !

Hoofdstuk 4

Verzamelstructuren

De ‘verzamelstructuren’ in Magma nemen een belangrijke plaats in onder de datastructuren
omdat ze de gebruiker in staat stellen een aantal veel voorkomende constructies uit te voeren
in één stap die anders omslachtiger in een aantal stappen zouden moeten gebeuren. In het
bijzonder is het mogelijk met verzamelingen en rijen in één stap dezelfde bewerking een groot
aantal keren uit te voeren en het resultaat op te slaan.

Een heel belangrijk principe bij de belangrijkste verzamelstructuren, die van verzamelingen
en rijtjes, is dat deze slechts objecten kunnen bevatten die element zijn van dezelfde wiskundige
structuur (groep, ring, enz.) en die dus van hetzelfde type zijn.

4.1 Verzamelingen

Verzamelingen in Magma zijn ongeordende collecties objecten uit één structuur zonder herha-
lingen.

De eenvoudigste manier om een verzameling te maken is door de elementen ervan op te
sommen en tussen accoladen { } te plaatsen: S := { 3, 7, 15 }; bijvoorbeeld. Wanneer
elementen nogmaals, of in een andere volgorde, in de verzameling worden gestopt verandert het
resultaat niet: { 15, 3, 15, 7, 3 } is niet te onderscheiden van de voorgaande verzameling
S. De elementen van de verzameling zijn dan ook niet geordend, en het is daarom niet mogelijk
te vragen om het i-de element van de verzameling. Wel kan met Representative of Random
een representatief of een willekeurig element uit de verzameling gehaald worden.

Voor gehele getallen in arithmetische progressie is de constructie { a .. b by c } nuttig,
die de getallen a, a + c, a + 2 · c enzovoorts tot (eventueel met) b in de verzameling stopt. De
getallen a, b, c mogen negatief zijn (maar b 6= 0).

De vereiste dat de elementen allemaal uit eenzelfde structuur moeten komen is een aanzien-
lijke beperking; deels wordt die opgevangen door Magma’s automatische coërcie, een mechanis-
me dat probeert voor elementen uit verschillende structuren een gemeenschappelijk structuur
te vinden waar beiden toe behoren; die structuur heet het universum, en is beschikbaar met
Universe. Dit gebeurt alleen wanneer dat universum welgedefinieerd en eenduidig is. Zo zal
Magma er geen moeite mee hebben om voor { 1, 1/2, 1.0 } achtereenvolgens als universum
Z, Q en uiteindelijk R te kiezen (na het invoegen van het gehele getal 1, de breuk 1/2 en het
reële getal 1).

Opgave 23. Hoeveel elementen denk je dat { 1, 1/2, 1.0 } zal hebben? Verifieer dat!

In zijn meest algemene vorm ziet de constructie van een verzameling er zo uit:

{ U | e(x) : x in S | P(x) },
hetgeen het volgende betekent: creëer de verzameling van elementen e(x), waar e een uitdruk-
king in (functie van) x is, voor alle x die in de eindige structuur S zitten en voldoen aan het
‘predicaat’ (de Boolese functie) P (x); stop tenslotte de resultaten e(x) allemaal in de structuur
U , die dan ook het universum van de verzameling wordt. Hier is een eenvoudig voorbeeld:

{ RealField() | (x-1)/(x+1) : x in [1..100 by 2] | IsPrime(x) }.
Merk op dat we hier de a..b by c constructie voor een rijtje [ ] (zie hieronder) hebben
gebruikt.

Een aantal ingrediënten mag weggelaten worden: wanneer U | er niet is zoekt Magma het
‘kleinste’ universum dat voldoet; het universum moet de eigenschap hebben dat alle elementen

17

e(x) er in ieder geval met U ! e(x) in gestopt kunnen worden. In het voorbeeld wordt
{ (x-1)/(x+1) : x in [1..100 by 2] | IsPrime(x) }

een verzameling rationale getallen. Ook kan het predicaat weggelaten worden (met de | ),
hetgeen in het voorbeeld natuurlijk een grotere verzameling oplevert:

{ (x-1)/(x+1) : x in [1..100 by 2] }.
Vaak zal de uitdrukking e(x) gewoon x zijn; in het voorbeeld komen we dan weer terug op de
oorspronkelijke constructie:

{ x : x in [1..100 by 2] }
en

{ 1..100 by 2 }.
zijn hetzelfde. De combinatie

{ RealField() | x : x in [1..100 by 2] }
is ook toegestaan.

Wordt alles weggelaten dan krijgen we de constructie
{ }

voor de lege verzameling; hiervan is het universum niet bepaald zolang we geen elementen in de
verzameling stoppen — het blijft de zogenaamde null-verzameling. Om een lege verzameling
met een gedefinieerd universum te maken kan

{ U | }
gebruikt worden. Tenslotte moet nog opgemerkt worden dat de uitdrukking e ook meer dan
één argument mag hebben; zo definieert

{ x1^2 + x2^2 : x1 in [0..9 by 2], x2 in [1..10 by 2] }

de verzameling van sommen van een kwadraat van een oneven getal en van een even getal
onder de tien.

4.2 Rijtjes

Een rijtje (sequence) in Magma is een geordende collectie objecten uit één wiskundige structuur,
waaronder herhalingen mogen voortkomen. Met S[i] kan het i-de element uit de rij gehaald
worden (waar de telling begint bij 1 en eindigt bij het aantal elementen van S). Het aantal
elementen van een rijtje (of verzameling) S is met #S te verkrijgen.

De constructie van rijtjes is analoog aan die van verzamelingen, zij het dat [ ] gebruikt
wordt in plaats van { }. Dus, in zijn meest algemene vorm

[ U | e(x) : x in S | P(x) ],
waar de objecten dezelfde betekenis hebben en restrikties kennen als boven.

Opgave 24. Creëer de verzameling van alle priemgetallen kleiner dan of gelijk aan 1003 die 5 modulo

8 zijn.

Opgave 25. Hoeveel priemgetallen kleiner dan 10000 zijn er? En kleiner dan 106?

Opgave 26. Bepaal de verzameling van alle getallen tussen 100 en 200 die door 11 deelbaar zijn op

minstens 2 verschillende manieren.

{ U | e(x) : x in S | P(x) } [ U | e(x) : x in S | P(x) ]

Opgave 27. Creëer de verzameling van getallen tussen 500 en 600 die niet deelbaar zijn door 2 en ook

niet door 7, maar wel door 11 of door 13.

Opgave 28. Hoeveel getallen zijn som van een kwadraat van een oneven getal en van een even getal

onder de tien? En hoeveel van die getallen worden meer dan eens gerepresenteerd in die vorm?

Opgave 29. De functie PrimeBasis geeft van een positief geheel getal het rijtje verschillende priem-

delers. Wat is het gemiddelde aantal priemdelers van de getallen tot 1000?

Opgave 30. Gebruik een Cartesisch product om alle paren (i, j) van gehele getallen die in absolute

waarde kleiner dan 10 zijn te vinden waarvoor het quotient i/j ongelijk aan 1 is.

4.2.1 Operaties op Verzamelingen en Rijtjes

Belangrijke binaire operaties op verzamelingen zijn join en meet voor vereniging en doorsne-
de, en diff en sdiff voor verschil en symmetrisch verschil. Om zulke (en sommige andere
operaties) uit te voeren is het soms nodig dat Magma een gezamenlijk universum bepaalt dat
de universa van de argumenten omvat. Met eq en ne kan vergeleken worden of twee verzame-
lingen gelijk of verschillend zijn, met U subset V of U ⊂ V en met u in U of u ∈ U . Ook
notin en notsubset bestaan. Voor kleine verzamelingen kan met Subsets de collectie van
alle deeelverzamelingen, of via Subsets(S, k) de collectie van alle deelverzamelingen van S
met precies k elementen, bepaald worden.

De voornaamste binaire operatie op rijtjes is die van concatenatie: s cat t plakt rijtje t
achter s mits er een gezamelijk uniersum gevonden wordt. Met in en notin kan weer getest
worden of een element al dan niet in een rijtje zit.

De predicaten IsEmpty en IsNull gaan na of een rij of verzameling leeg, danwel leeg zonder
universum, is. Met # wordt het aantal elementen gevonden. Het paar Seqset en Setseq (voor
SequenceToSet en SetToSequence maakt van een rij een verzameling en omgekeerd.

Wanneer de elementen van een rij of verzameling uit een geordende structuur komen (waar-
op een < relatie is gedefinieerd), dan kan met de functies Max en Min het maximum en minimum
van de rij of verzameling bepaald worden, met Sort kan de rij gesorteerd worden. Van Sort

zijn twee versies beschikbaar, namelijk een functie en een procedure. Het verschil tussen beide
in Magma is dat een functie output levert in de vorm van een object dat wordt afgeleverd en een
procedure doet dat niet; bovendien kan een procedure sommige elementen die als input worden
meegeleverd wijzigen, terwijl een functie nooit argumenten kan veranderen. In dit geval, als
S een rijtje is kun je met T := Sort(S); de gesorteerde versie van S in T terugkrijgen; als
procedure kun je Sort(~S); typen waarop Magma het object S verandert in een gesorteerde
versie. Voor hele grote rijen kan het belangrijk zijn dat voor de procedurele versie maar 1
rij blijft bestaan, terwijl bij de functie-aanroep zowel de oorspronkelijke S als de gesorteerde
variant aanwezig zijn.

join meet diff sdiff eq ne in notin subset Subsets Seqset Setseq Sort Max Min ~

Om rijtjes en verzamelingen uit te breiden zijn er functies en prcocedures als Append

(plakt een element achter een bestaande rij) en Include (stopt een element in een bestaande
verzameling). Van Append zijn er weer een procedurele versie gebruiken met Append(~S, x);

en een functie T := Append(S, x);. Net zo voor Include voor een verzameling. Ook bestaat
Include voor rijtjes, waar alleen het element achter de rij geplakt wordt wanneer het niet
al in de rij zit. Om het i-de element uit een rij te verwijderen gebruike men Remove(~S,

i) of Remove(S, i). Met Exclude(~S, x) wordt (de eerste) x uit de verzameling (of rij) S
verwijderd; een rij wordt daar korter van (omdat de andere elementen opschuiven).

Het is ook mogelijk de waarde van het i-de element van een rijtje s te veranderen met de
assignment S[i] := y; De inhoud wordt dan overschreven. Omdat i groter mag zijn dan de
lengte van het rijtje, kan de rij zo ook verlengd worden; bovendien is het mogelijk ‘gaten’ in
de rij te laten vallen.

Opgave 31. Hoe ziet s er uit na s := [1,2]; Append(~s,3); en s[#s+2] := 5; en s[#s+2] := 8;?

Opgave 32. Laat zien dat de verzamelingen {1, 2, 3, 4} en {1, 3, 2, 4} in Magma hetzelfde zijn, maar

dat de rijtjes [1, 2, 3, 4] en [1, 3, 2, 4] verschillend zijn.

Opgave 33. Creëer van de priemgetallen P tot 1000 de deelrij Q van elementen waarvoor p−4 of p+4

weer een priemgetal is. Is R = {863, 911} een deelverzameling van Q?

Opgave 34. Bepaal de rij van laatste cijfers van de getallen in Q (uit de vorige opgave); bepaal ook

een rijtje paren [c, f ] waar c door de voorkomende laatste loopt en f het aantal malen dat het optreedt

in Q.

Opgave 35. Bepaal de collectie van priemgetallen die optreden als deler van een getal van de vorm

q − 1, met q zelf ook een priemgetal.

Opgave 36. De functie σi bepaalt van een natuurlijk getal de som van de i-de machten van de delers

van n. Je vind het resultaat in Magma met DivisorSigma(i, n). Gebruik dat om een verzameling S

van getallen onder de 10000 te vinden met precies 48 delers. Bepaal ook de doorsnede hiervan met de

getallen die precies 3 priemdelers hebben.

Opgave 37. Vind een voorbeeld van een rijtje waar R waar Setseq(Seqset(R)) niet gelijk is aan R.

Kun je ook een voorbeeld van een verzameling V vinden waarvoor Seqset(Setseq(V)) niet gelijk is

aan V ?

4.3 Overige Verzamelstructuren

Van de overige verzamelstructuren noemen we hier alleen maar de geı̈ndiceerde verzamelin-
gen waar de elementen wél geordend zijn, te maken met {@ @}; de multiverzamelingen waar
elementen met multipliciteiten kunnen voorkomen, te maken met {* *}; de formele verzame-

lingen die weinig meer zijn dan predicaten, te maken met {! !}; de tuples (elementen van
een Cartesisch product), te maken met < > en de lists, met [* *]. Voor de laatste twee gelden
niet de strenge eisen op het universum, zodat er geheel verschillendsoortige objecten in mogen
zitten.

Append Include Exclude Remove {@ @} {* *} {! !} [* *] < >


